

# Temporal Logic, $\mu$ Kanren, and a Time-Traveling RDF Database

NATHANIEL RUDAVSKY-BRODY\*, Independent Researcher, USA

By adding a temporal primitive to the  $\mu$ Kanren language and adapting its interleaving search strategy to preserve simultaneity in linear time, we show how a system of temporal relational programming can be implemented. This system is then applied to a practical problem in distributed systems and linked data, as we describe a simple data store that can provide incremental solutions (deltas) to complex queries as new records are added to, or removed from the database.

Additional Key Words and Phrases: miniKanren, relational programming, temporal logic, RDF, microservices

## 1 INTRODUCTION

When using logic programming to model or interact with stateful resources, it is convenient to have a way to reason about time. To this end, temporal logic, and in particular linear temporal logic, has been formalized as a modal extension to predicate logic. Most definitions of linear temporal logic begin with two modal operators, Next (X) and Until (U), which are used to construct a larger set of operators for conveniently reasoning about time. Informally speaking, for the formulas  $a$  and  $b$  we say that  $X\ b$  holds if  $b$  is true in the next time step, and  $a\ U\ b$  holds if  $a$  is always true until  $b$  is true, and  $b$  is eventually true. For a good review of this work, see [Orgun and Ma 1994].

A number of implementations of linear temporal logic have been proposed as extensions to logic languages such as Prolog. This paper stems from the simple insight that an analogous modeling of linear time can be achieved in the miniKanren family of languages [Friedman et al. 2018], a shallowly embedded logic programming DSL with interleaving depth-first search. Adding a single temporal primitive and adapting the interleaving search strategy to account for the concepts of ‘now’ (simultaneity) and ‘later’ allows us to build tools for temporal reasoning that turn out to integrate quite well with the basic methods of relational programming. The new temporal primitive provides a way of controlling simultaneity and goal construction in miniKanren programs, and be extended in a number of ways that are interesting from both a theoretical and practical standpoint.

This paper is organized in three parts. In the first two sections we go through the low-level implementation, using as our starting point the minimalist  $\mu$ Kanren language. Then, we sketch out some preliminary ideas on how this operator might be extended to support a more complete linear temporal logic. Finally, we turn to a real-world application in the domain of distributed systems and linked data, and see how the same low-level operator can be used to build a time-aware accessor to a stateful data structure leveraging miniKanren’s interleaving search.<sup>1</sup>

The motivation for the last example is as follows. In distributed systems such as a microservice architecture<sup>2</sup> we often want to send push updates based on *deltas*, or entries that have been added

\*Parts of the research for this paper were undertaken while the author was employed as Semantic Applications Developer at Tenforce (Leuven, Belgium). The author thanks Tenforce for its support in undertaking this research. Any opinions, findings, and conclusions or recommendations expressed in this material are those of the author and do not necessarily reflect the views of Tenforce.

<sup>1</sup>The full code for this paper is available at <https://github.com/nathanielrb/time-in-microKanren>.

<sup>2</sup>For the specific context, see [Versteden and Pauwels 2016] and [Versteden and Pauwels 2018].

Author’s address: Nathaniel Rudavsky-Brody, Independent Researcher, Sebastopol, California, 95472, USA, [nathaniel@quince.studio](mailto:nathaniel@quince.studio).

2018. 2475-1421/2018/1-ART1 \$15.00

<https://doi.org/>

to or removed from the database. In practice, this often means having a service that indiscriminately pushes all deltas to interested subscribers; it is then the responsibility of each subscriber to filter the deltas and determine which, if any, are relevant to its own operations. If we can calculate deltas to specific queries, however, this whole process can be greatly refined. Ultimately we will describe how a simple RDF database (triple store) can be implemented using temporal  $\mu$ Kanren as its query language, that calculates deltas to specific queries via a system of incremental indexes.

## 2 BASIC DEFINITIONS

We begin by briefly reviewing the definitions of  $\mu$ Kanren as described in [Hemann and Friedman 2013] and [Hemann et al. 2016], deferring to those articles for discussion and motivations. To better expose the core concepts, we choose to work with the earlier implementation<sup>3</sup> which has the advantage of using procedures to represent immature streams, leaving us free to use Scheme promises for time. Readers familiar with its implementation may wish to skip to the next section. Though the two versions of  $\mu$ Kanren run to 39 and 54 lines of code respectively, here we will only repeat the definitions from [Hemann and Friedman 2013] relevant to understanding the changes that will be introduced afterwards. (The omitted definitions can be referred to in the final implementation included in Appendix A.)

A  $\mu$ Kanren program operates by applying a *goal*, analogous to a predicate in logic programming and implemented as a procedure of one argument, to a *state*, defined as a substitution (an association list of variables and their values) paired with a variable counter. Variables are represented as vectors containing their variable index. The program can succeed or fail, and when it succeeds returns a sequence of new states, called a *stream*, each one extending the original state by new variable substitutions that make the goal succeed.

Goals are built with four basic goal constructors, `==`, `call/fresh`, `disj` and `conj`, to be defined below. In the following example, a goal made of the disjunction of two `==` goals is applied to the empty state. The program returns a stream of two states, each corresponding to one branch of the disjunction.

```
(define empty-state '(() . 0))

((call/fresh
  (lambda (q)
    (disj (== q 4)
          (== q 5))))
 empty-state)
;; => '(((#(0) . 4)) . 1) (((#(0) . 5)) . 1))
```

The walk operator, not defined here, searches for a variable's value in a substitution, while substitutions are extended (without checking for circularities) by consing on the new substitution. `walk` is used by the `unify` operator, which recursively unifies two variables with respect to a given variable substitution, extending the substitution when it succeeds, and returning `#f` when it fails. It is `unify`'s behavior that defines the basic terms of  $\mu$ Kanren as being variables, objects equivalent under `eq?`, and pairs of such terms.

The fundamental goal constructor is `==`, which builds a goal that succeeds if its two arguments can be unified in the current state, and otherwise returns the empty stream (`mzero`).

<sup>3</sup>See <https://github.com/jasonhemann/microKanren> for the code.

```

99 (define (== u v)
100   (lambda (s/c)
101     (let ((s (unify u v (car s/c))))
102       (if s (unit `(,s . ,(cdr s/c))) mzero))))
103
104   The conj and disj goal constructors return goals that succeed if, respectively, both or either
105   of the goals passed as arguments succeed. They are defined in terms of mplus and bind, which
106   together implement  $\mu$ Kanren's interleaving search strategy.
107   (define (disj g1 g2) (lambda (s/c) (mplus (g1 s/c) (g2 s/c))))
108   (define (conj g1 g2) (lambda (s/c) (bind (g1 s/c) g2)))
109
110   (define (mplus $1 $2)
111     (cond
112       ((null? $1) $2)
113       ((procedure? $1) (lambda () (mplus $2 ($1)))) ; interleave
114       (else (cons (car $1) (mplus (cdr $1) $2)))))
115
116   (define (bind $ g)
117     (cond
118       ((null? $) mzero)
119       ((procedure? $) (lambda () (bind ($) g)))
120       (else (mplus (g (car $)) (bind (cdr $) g)))))

```

The interleaving itself, which guaranties complete search without the performance penalty of breadth-first search, is effectuated by the second case of mplus by exchanging the order of \$1 and \$2. Moreover, the second case in each of the two operators implements *immature streams* as lambda expressions, allowing programs to return infinite streams of results. In  $\mu$ Kanren it is the user's responsibility to correctly handle immature streams, invoking them if necessary. A regular stream, namely a pair of a state and a stream, is correspondingly termed a *mature stream*.

### 3 TIME IN $\mu$ KANREN

To model linear time in  $\mu$ Kanren we begin by introducing a third type of stream, *delayed streams*, to represent goals that are delayed until a later point in time. We can represent delayed streams by promises, and construct them using a single temporal goal constructor next analogous to the Next (X) operator introduced above.

```

132 (define-syntax next
133   (syntax-rules ()
134     ((_ g) (lambda (s/c) (delay (g s/c))))))
135

```

A complex goal might have many levels of delays representing different future points in time. It is important that the temporal order of these subsidiary goals be preserved during interleaving search, both with regards to the ordering of solutions and also to guaranty that a goal (which might refer to a stateful resource) is constructed at the right time. We accomplish this by adapting the interleaving process through the addition of two cases to the definition of mplus, so that delayed goals (promises) are shunted right, all promises at the same nested level are recombined together. Among other properties, this preserves the order-independence of conjunction and disjunction.

```

148 (define (mplus $1 $2)
149   (cond
150     ((null? $1) $2)
151     ((procedure? $1) (lambda () (mplus $2 ($1))))
152     ((and (promise? $1) (promise? $2))
153      (delay (mplus (force $1) (force $2)))) ; recombine
154     ((promise? $1) (mplus $2 $1)) ; shunt right
155     (else (cons (car $1) (mplus (cdr $1) $2)))))

```

bind also needs to be modified by adding an additional case to delay the binding of delayed streams with a goal. The utility function forward is used to ‘fast-forward’ through the enclosing delayed stream.

```

159 (define (forward g)
160   (lambda (s/c)
161     (let rec ((g s/c))
162       (cond ((null? $) '())
163             ((promise? $) (force $))
164             ((procedure? $) (lambda () (rec ($))))
165             (else (cons (car $) (rec (cdr $)))))))
166
167 (define (bind $ g)
168   (cond
169     ((null? $) mzero)
170     ((procedure? $) (lambda () (bind ($) g)))
171     ((promise? $) (delay (bind (force $) (forward g))))
172     (else (mplus (g (car $)) (bind (cdr $) g)))))

```

The resulting 53 lines of code make up the functional core of our temporal  $\mu$ Kanren. To keep the subsequent code simple, we extend the miniKanren wrappers from the original paper to handle delayed streams, saving the code for Appendix B. Using the miniKanren wrappers, and in particular run\* instead of call/fresh to reify solutions, improves the readability of  $\mu$ Kanren’s data representation. We also introduce a few convenience functions for accessing delayed streams, here defined with the help of take-right and drop-right from SRFI-1.

```

180 (define (promised $)
181   (take-right $ 0))
182
183 (define (current $)
184   (drop-right $ 0))
185
186 (define (advance $)
187   (let ((p (promised $)))
188     (and p (force p))))

```

We are ready for a simple example. Here we use disj to specify that q equals \*db\* either now or at the next point in time. The first solution in the returned stream is none other than 1, the current value of \*db\*, while the promise contains the delayed stream to be advanced at a future point in time.

```

194 (define *db* 1)

```

```

197 (define r
198   (run* (q)
199     (fresh (a b)
200       (disj (== q *db*)
201         (next (== q *db*))))))
202 ;; => (1 . #<promise>)

```

Now we can update our database and advance `r` to continue forward in time. At this point the stream is fully mature, since `next` is not used recursively.

```

206 (set! *db* 2)
207
208 (advance r)
209 ;; => (2)

```

The shunt-right mechanism turns out to have interesting properties for interleaving search even without referring to a stateful resource, for instance in ordering and grouping solutions, as the following contrived example shows. We start by defining `inco` which recursively generates a ‘now or later’ type of goal, incrementing its variable at each future point in time.

```

215 (define (inco x)
216   (let rec ((n 0))
217     (disj (== x n)
218       (next (rec (+ n 1))))))

```

The program `s` uses `inco` to generate pairs of such incrementing variables.

```

221 (define s
222   (run* (q)
223     (fresh (a b)
224       (== q `(,a ,b))
225       (conj (inco a) (inco b)))))

```

As we advance `s`, we see that the solutions are grouped in increasing order of the maximum value of the two variables.

```

228 s
229 ;; => ((0 0) . #<promise>))
230
231 (advance s)
232 ;; => ((0 1) (1 0) (1 1) . #<promise>))
233
234 (advance (advance s))
235 ;; => ((1 2) (2 0) (2 1) (2 2) (0 2) . #<promise>))
236
237 (advance (advance (advance s)))
238 ;; => ((0 3) (2 3) (3 0) (3 1) (3 2) (3 3) (1 3) . #<promise>))
239

```

Indeed, this points up the difference between two basic uses, or two interpretations, of temporal  $\mu$ Kanren. In the first case, when interacting with a stateful resource, we are doing logic programming *in* time. When reasoning *about* time in a truly relational way, on the other hand, we are more interested in the combinatoric properties of `next`. We will scratch the surface of this distinction in the next section, but a deeper consideration of its implications remains to be done.

#### 4 TOWARDS A TEMPORAL RELATIONAL PROGRAMMING

As we saw in the introduction, linear temporal logic extends predicate logic with temporal modal operators referring to linear time. Several of these operators can be usefully recovered in temporal  $\mu$ Kanren using the `next` primitive defined above. We will not develop a formally complete system here, but rather provide some heuristic examples to suggest how such a system might developed, noting along the way its potential limitations and specificities.

Most temporal operators need to be defined recursively. When doing this, it is essential to control when the goal is constructed, since the goal might refer to a stateful resource. For temporal  $\mu$ Kanren we can do this by wrapping goals in a lambda expression which are evaluated at the appropriate time. The definition of `precedes`, modeled on Weak Until (W) and which stipulates that a goal `g` holds at least until a second goal `h` holds, shows how this can be done.

```
(define (precedes* g* h*)
  (let ((g (g*)) (h (h*)))
    (disj h (conj g (next (precedes* g* h*)))))
```

```
(define-syntax precedes
  (syntax-rules ()
    ((_ g h) (precedes* (lambda () g) (lambda () h)))))
```

Translating another basic operator, `Always (G)`, reveals one of the pitfalls of naively mapping temporal logic to relational programming. Our first impulse is to define it analogously to `until`.

```
(define (always* g*)
  (let ((g (g*)))
    (conj g (next (always* g*)))))
```

```
(define-syntax always
  (syntax-rules ()
    ((_ g) (always* (lambda () g)))))
```

A little experimentation, however, shows how this definition can be problematic. A goal constructed with `always` will return a promise so long as the goal it encloses holds, and the empty stream as soon as it fails. This behavior can be useful for a program that wants to verify whether a state still holds, but `always` will never construct a mature stream of solutions, since clearly the goal will never have been true *forever*, at least not in linear time.

There are two ways of addressing this limitation, depending on how we want to use our operator. One is to modify the definition of `next` in such a way as to allow for an `eot` (end-of-time) accessor that would essentially cut off the forward recursion of delayed promises. Another approach is to define the weaker and impure `as-long-as`, which constructs a stream of solutions to the goal `h` that continues as long as the goal `g` still holds. This technique of writing ‘guarded’ goal constructors is reminiscent of the implementation of cuts in `miniKanren`.

```
(define (as-long-as* g* h*)
  (let ((g (g*)) (h (h*)))
    (lambda (s/c)
      (let (($ (g s/c)))
        (if (null? $) mzero
            (bind $ (disj h (next (as-long-as* g* h*)))))
```

```

295 (define-syntax as-long-as
296   (syntax-rules ()
297     ((_ g h) (as-long-as* (lambda () g) (lambda () h))))))

```

The Eventually (F) operator also has many useful applications, but presents a similar caveat regarding infinite time. Still, the following naive definition will be useful in some contexts.

```

300 (define (eventually* g*)
301   (let ((g (g*)))
302     (disj g (next (eventually* g*)))))
303
304 (define-syntax eventually
305   (syntax-rules ()
306     ((_ g) (eventually* (lambda () g)))))

```

The guarded definition, however, provides more intuitive behavior, by cutting off further recursion when the goal produces a non-empty stream of solutions.

```

310 (define (eventually* g*)
311   (let ((g (g*)))
312     (lambda (s/c)
313       (let (($ (g s/c)))
314         (if (null? $)
315             (bind (list s/c) (next (eventually* g*)))
316             $))))))

```

```

318 (define-syntax eventually
319   (syntax-rules ()
320     ((_ g) (eventually* (lambda () g)))))

```

Either one can be used to define an analogue of strong Until, which along with Next is one of the building blocks of linear temporal logic.

```

324 (define-syntax until
325   (syntax-rules ()
326     ((_ g h) (conj (precedes g h) (eventually h)))))

```

Clearly this is only a beginning, and as the above examples suggest, what constitutes a *useful* temporal relational programming language will be determined in part by the application domain. Still, developing a formally satisfying version of such a system would be an obvious next step. All this side-stepping of infinite time also begs the question of what it really means to model it, a question we will not explore further here, since our primary focus is on using miniKanren to interact with stateful resources in a temporally-aware way. Giving a satisfactory answer to this question could be a fruitful direction for further work.

Another would be recovering temporal  $\mu$ Kanren's functionality in a version of miniKanren with constraints, namely cKanren [Alvis et al. 2011] or the extended  $\mu$ Kanren described in [Hemann and Friedman 2015]. It would also be interesting to explore the integration of temporal logic with different representations of negation in miniKanren that have been pursued in both the published and unpublished literature.

## 5 CALCULATING DELTAS WITH INCREMENTAL SEARCH

Now we turn to the practical application described in the introduction, and see how our temporal primitive next can be used to implement a simple data store with temporally-aware incremental

search. We begin by describing the implementation, and then present the full code in the following section.

Our goal is to design an RDF database (triple store) using temporal  $\mu$ Kanren as its query language, that calculates new or invalidated solutions to specific queries as records are updated in the database. By using the next constructor and a simple system of incremental indexes, we can store the final search positions for a query. Running a query will return both the current results and a delayed stream that when advanced will continue searching at the previous search-tree's leaves. Therefore, advancing the delayed stream after the database has been updated will return solutions that have been added to, or subtracted from the solution set.

An RDF database [Manola and Miller 2004] stores semantic facts, or triples,<sup>4</sup> made up of a *subject* (a URI), a *predicate* (a URI), and an *object* (a URI, or a string, boolean or numeric literal).

```
<http://ex.org/people/1> <http://xmlns.com/foaf/0.1/name> "John"
```

A minimalist triple store can be implemented using three indexes, allowing for the retrieval of groups of triples matching a given query pattern: spo, pos, and osp, where s is the subject, p the predicate, and o the object. For the query pattern ?s <P> <O>, for instance, ?s is our only variable so the database would use the pos index for fastest lookup. Here, however, we want to know *incremental indexes*: for a given s, we need the indexed ps in a form that can be easily compared to future ps to determine whether new keys have been added.

Our database is therefore defined as a set of seven incremental indexes (null, s, sp, p, po, o, os) plus the spo index for full triples. Adding a triple to the database is a matter of consing each element to the appropriate incrementals lists and adding a truthy value (here #t, though this could also be a more meaningful identifier or timestamp) to the full spo index. When deleting a triple, we leave the index keys, but update the triple's value in spo to #f. Here are the incremental indexes after adding the triple <A> <B> <C>.

```
<∅ [ #f => (<A> ...) ] >
<s [ <A> => (<B> ...) ] >
<sp [ (<A> <B>) => (<C> ...) ] >
<p [ <B> => (<C> ...) ] >
<po [ (<B> <C>) => (<A> ...) ] >
<o [ <C> => (<A> ...) ] >
<os [ (<C> <A>) => (<B> ...) ] >
<spo [ (<A> <B> <C>) => #t ] >
```

The main accessor is the goal constructor triple-nolo ('triple-now-or-later') that descends recursively through the incremental indexes one level at a time, constructing a stream with the current indexes and saving the last search positions in a delayed stream. A dynamic parameter is used to specify the current database state.

To see our data store in action, we consider the following query written in SPARQL 1.1 [Harris and Seaborne 2013], the standard query language for RDF data stores. This query will be translated into temporal  $\mu$ Kanren and run against successive states of the database.

```
SELECT ?o
WHERE {
  <S> <P> ?o.
  <Q> <R> ?o.
}
```

First, however, we need to define an empty database.

<sup>4</sup>Here we will only consider triples, though triple stores actually store quads, with the addition of the *graph* element.



```
393 (define db0 (empty-db))
```

394 Then the SPARQL query is translated into the temporal  $\mu$ Kanren goal `r` using `triple-nolo`. To  
 395 make it clear what is going on, we keep track of the individual delta flags for each triple goal, though  
 396 in practice we usually want to know only whether the solution was added (all `+`s) or removed (at  
 397 least one `-`). Since there are no solutions to our query, the program returns a delayed stream.

```
398 (define r
399   (parameterize ((latest-db db0))
400     (run* (q)
401       (fresh (o deltas d1 d2)
402         (== q `(:,deltas ,o))
403         (== deltas `(:,d1 ,d2))
404         (triple-nolo d1 '<S> '<P> o)
405         (triple-nolo d2 '<Q> '<R> o))))))
406 ;; => #<promise>
```

407 Now we can add some triples and advance the delayed stream.

```
408 (define db1
409   (add-triples db0 '(((<S> <P> <01>)
410                      (<S> <P> <02>)
411                      (<Q> <R> <01>)
412                      (<A> <B> <C>))))))
413
414
```

```
415 (parameterize ((latest-db db1))
416   (advance r))
417 ;; => (((+ +) <01>) . #<promise>)
```

418 If we delete a triple that contributed to one of our solutions, that solution will be returned with a  
 419 negative delta flag in the next iteration.

```
420 (define db2
421   (delete-triples db1 '(((<S> <P> <01>))))))
422
```

```
423 (parameterize ((latest-db db2))
424   (advance (advance r)))
425 ;; => (((+ -) <01>) . #<promise>)
```

426 Finally we add that triple back along with some new solutions, to get positive deltas.

```
427 (define db3
428   (add-triples db2 '(((<S> <P> <01>)
429                      (<S> <P> <03>)
430                      (<Q> <R> <03>)
431                      (<S> <P> <M>)
432                      (<Q> <R> <M>))))))
433
434
```

```
435 (parameterize ((latest-db db3))
436   (advance (advance (advance r))))
437 ;; => (((+ +) <01>) ((+ +) <M>) ((+ +) <03>) . #<promise>)
```

438 Here we have been proceeding linearly, but since the database states are persistent we can  
 439 calculate deltas over any two states. Keeping track of a stream of states and indexing them on time  
 440 will therefore give us a truly time-traveling data store.

## 6 IMPLEMENTING THE RDF STORE

Now we are ready to walk through the full code for the data store.

We start by defining `project`, a standard miniKanren operator [Byrd 2009] that binds a set of variables to their current substitutions. It will be used control triple-nolo's recursive descent through the indexes.

```
(define-syntax project
  (syntax-rules ()
    ((_ (x ...) g)
     (lambda (s/c)
       (let ((x (walk x (car s/c))) ...)
         (g s/c))))))
```

A database is defined as a set of seven incremental indexes plus the full `spo` index, as described above. We keep a separate incremental index for each index level, storing the incrementals as a simple cons list along with a map for quick existence checking. For each index itself we use a hash array mapped trie<sup>5</sup> that allows for persistently storing successive states of the database through path copying.

```
(define-record db null s sp p po o os spo)
```

```
(define-record incrementals map list)
```

```
(define (empty-incrementals)
  (make-incrementals (persistent-map) '()))
```

```
(define (empty-db)
  (apply make-db
         (make-list 8 (persistent-map))))
```

Dynamic parameters are used to specify the current database state.

```
(define latest-db (make-parameter (empty-db)))
```

```
(define (latest-incrementals accessor key)
  (incrementals-list
   (map-ref (accessor (latest-db)) key (empty-incrementals))))
```

```
(define (latest-triple s p o)
  (map-ref (db-spo (latest-db))
           (list s p o)))
```

To update the indexes we update the quick-check map to `#t` and cons the new value to the incrementals list.

<sup>5</sup>In the Chicken implementation presented here, we use Persistent Hash Maps [Heidkamp 2013], derived from Ian Price's Hash Array Mapped Tries [Price 2014].

```

491 (define (update-incrementals table key val)
492   (let ((incrementals (map-ref table key (empty-incrementals))))
493     (map-add table key
494       (if (map-ref (incrementals-map incrementals) val)
495         incrementals
496         (make-incrementals
497           (map-add (incrementals-map incrementals) val #t)
498           (cons val (incrementals-list incrementals)))))))
499

```

```

500 (define (update-triple table triple val)
501   (map-add table triple val))

```

Updating (adding or removing) a triple is therefore a matter of updating all eight indexes.

```

502 (define (update-triples DB triples val)
503   (let loop ((triples triples)
504             (i/null (db-null DB))
505             (i/s (db-s DB))
506             (i/sp (db-sp DB))
507             (i/p (db-p DB))
508             (i/po (db-po DB))
509             (i/o (db-o DB))
510             (i/os (db-os DB))
511             (i/spo (db-spo DB)))
512     (if (null? triples)
513       (make-db i/null i/s i/sp i/p i/po i/o i/os i/spo)
514       (match (car triples)
515         ((s p o)
516          (loop (cdr triples)
517                (update-incrementals i/null #f s)
518                (update-incrementals i/s s p)
519                (update-incrementals i/sp (list s p) o)
520                (update-incrementals i/p p o)
521                (update-incrementals i/po (list p o) s)
522                (update-incrementals i/o o s)
523                (update-incrementals i/os (list o s) p)
524                (update-triple i/spo (list s p o) val)))))))
525

```

```

526 (define (add-triples DB triples) (update-triples DB triples #t))

```

```

527 (define (delete-triples DB triples) (update-triples DB triples #f))

```

```

528 (define (add-triple s p o)
529   (add-triples `((,s ,p ,o))))

```

```

530 (define (delete-triple s p o)
531   (delete-triples `((,s ,p ,o))))

```

At last we can define our main accessor, `triple-nolo`. This operator descends recursively through the indexes, selecting which index to use according to which of its arguments are variables

(var?) and using project in the recursion step to reduce the number of variables by one. The stream of solutions is constructed by stream that recurses through the incrementals list (indexes), starting at the head and stopping when it reaches the previous head (ref), and storing the new head of the list in another recursive call wrapped in next as the delayed stream of future solutions.

```
(define (triple-nolo delta s p o)
  (let ((mkstrm (lambda (var accessor key)
                  (let* ((get-incrementals (lambda ()
                                              (latest-incrementals accessor key)))
                         (initial-indexes (get-incrementals)))
                    (let stream ((indexes initial-indexes)
                                (ref '())
                                (next-ref initial-indexes))
                      (if (equal? indexes ref)
                          (next
                           (let ((vals (get-incrementals)))
                             (stream vals next-ref vals)))
                          (disj
                           (conj (= var (car indexes))
                                (project (s p o)
                                         (triple-nolo delta s p o)))
                           (stream (cdr indexes) ref next-ref)))))))
    (cond ((and (var? s) (var? p) (var? o)) (mkstrm s db-null #f))
          ((and (var? s) (var? p)) (mkstrm s db-o o))
          ((and (var? s) (var? o)) (mkstrm o db-p p))
          ((and (var? p) (var? o)) (mkstrm p db-s s))
          ((var? s) (mkstrm s db-po (list p o)))
          ((var? p) (mkstrm p db-os (list o s)))
          ((var? o) (mkstrm o db-sp (list s p)))
          (else
           (let leaf ((ref #f))
             (let ((v (latest-triple s p o)))
               (cond ((eq? v ref) (next (leaf v)))
                     (v (disj (= delta '+) (next (leaf v)))
                              (else (disj (= delta '-') (next (leaf v)))))))))))
```

Though this reduced example is clearly far from a production-ready system, we hope we have demonstrated one of the practical uses of temporal relational programming. Indeed, we intend to pursue these tools in an industrial setting, and the full implementation will be a laboratory for other practical applications of miniKanren, such as drawing on [Byrd et al. 2012] to compile SPARQL to miniKanren and using search ordering as explored in [Swords and Friedman 2013] to aid with query optimization.

## A APPENDIX: TEMPORAL $\mu$ KANREN

```
(define (var c) (vector c))
(define (var? x) (vector? x))
(define (var=? x1 x2) (= (vector-ref x1 0) (vector-ref x2 0)))

(define (walk u s)
```

```

589      (let ((pr (and (var? u) (assp (lambda (v) (var=? u v)) s))))
590        (if pr (walk (cdr pr) s) u)))
591
592 (define (ext-s x v s) `((,x . ,v) . ,s))
593
594 (define (= u v)
595   (lambda (s/c)
596     (let ((s (unify u v (car s/c))))
597       (if s (unit `(,s . ,(cdr s/c)) mzero))))
598
599 (define (unit s/c) (cons s/c mzero))
600 (define mzero '())
601
602 (define (unify u v s)
603   (let ((u (walk u s)) (v (walk v s)))
604     (cond
605       ((and (var? u) (var? v) (var=? u v)) s)
606       ((var? u) (ext-s u v s))
607       ((var? v) (ext-s v u s))
608       ((and (pair? u) (pair? v))
609        (let ((s (unify (car u) (car v) s)))
610          (and s (unify (cdr u) (cdr v) s))))
611       (else (and (eqv? u v) s))))))
612
613 (define (call/fresh f)
614   (lambda (s/c)
615     (let ((c (cdr s/c)))
616       ((f (var c)) `(,(car s/c) . ,(+ c 1)))))
617
618 (define (disj g1 g2) (lambda (s/c) (mplus (g1 s/c) (g2 s/c))))
619 (define (conj g1 g2) (lambda (s/c) (bind (g1 s/c) g2)))
620
621 (define (mplus $1 $2)
622   (cond
623     ((null? $1) $2)
624     ((procedure? $1) (lambda () (mplus $2 ($1))))
625     ((and (promise? $1) (promise? $2))
626      (delay (mplus (force $1) (force $2))))
627     ((promise? $1) (mplus $2 $1))
628     (else (cons (car $1) (mplus (cdr $1) $2)))))
629
630 (define (forward g)
631   (lambda (s/c)
632     (let rec ((g s/c))
633       (cond ((null? $) '())
634             ((promise? $) (force $))
635             ((procedure? $) (lambda () (rec ($))))
636             (else (cons (car $) (rec (cdr $)))))))
637

```

```

638 (define (bind $ g)
639   (cond
640     ((null? $) mzero)
641     ((procedure? $) (lambda () (bind ($) g)))
642     ((promise? $) (delay (bind (force $) (forward g))))
643     (else (mplus (g (car $)) (bind (cdr $) g)))))
644
645 (define-syntax next
646   (syntax-rules ()
647     ((_ g) (lambda (s/c) (delay (g s/c))))))
648

```

## 649 B APPENDIX: MINIKANREN WRAPPERS

650 Here we adapt the miniKanren control operators described in [Hemann and Friedman 2013] for  
 651 delayed streams. The main changes are to the definitions of run, run\*, pull, take-all and take.  
 652 Unlike the original paper, we do not use Zzz to wrap goals in conj+ and disj+, since this makes it  
 653 difficult to control the goal construction time at different levels of nested delays, a problem when  
 654 referring to stateful resources. This difficulty should be addressed in future implementations of  
 655 temporal  $\mu$ Kanren.

```

656
657 (define-syntax conj+
658   (syntax-rules ()
659     ((_ g) g)
660     ((_ g0 g ...) (conj g0 (conj+ g ...)))))
661
662 (define-syntax disj+
663   (syntax-rules ()
664     ((_ g) g)
665     ((_ g0 g ...) (disj g0 (disj+ g ...)))))
666
667 (define-syntax fresh
668   (syntax-rules ()
669     ((_ () g0 g ...) (conj+ g0 g ...))
670     ((_ (x0 x ...) g0 g ...)
671       (call/fresh
672         (lambda (x0)
673           (fresh (x ...) g0 g ...))))))
674
675 (define-syntax conde
676   (syntax-rules ()
677     ((_ (g0 g ...) ...) (disj+ (conj+ g0 g ...) ...))))
678
679 (define-syntax run
680   (syntax-rules ()
681     ((_ n (x ...) g0 g ...)
682       (let r ((k n) ($ (take n (call/goal (fresh (x ...) g0 g ...)))))
683         (cond ((null? $) '())
684               ((promise? $) (delay (r (- k 1) (take k (force $)))))
685               (else (cons (reify-1st (car $))
686                           (r (- k 1) (take k (force $)))))))
686

```

```

687         (r (- k 1) (cdr $)))))))))
688
689 (define-syntax run*
690   (syntax-rules ()
691     ((_ (x ...) g0 g ...)
692       (let r (($ (take-all (call/goal (fresh (x ...) g0 g ...))))))
693         (cond ((null? $) '())
694               ((promise? $) (delay (r (take-all (force $)))))
695               (else (cons (reify-1st (car $))
696                           (r (cdr $)))))))
697
698 (define empty-state '(() . 0))
699
700 (define (call/goal g) (g empty-state))
701
702 (define (pull $)
703   (cond ((procedure? $) (pull ($)))
704         ((promise? $) $)
705         (else $)))
706
707 (define (take-all $)
708   (let (($ (pull $)))
709     (cond ((null? $) '())
710           ((promise? $) $)
711           (else (cons (car $) (take-all (cdr $))))))
712
713 (define (take n $)
714   (if (zero? n) '()
715       (let (($ (pull $)))
716         (cond ((null? $) '())
717               ((promise? $) $)
718               (else (cons (car $) (take (- n 1) (cdr $))))))
719
720 (define (reify-1st s/c)
721   (let ((v (walk* (var 0) (car s/c))))
722     (walk* v (reify-s v '()))))
723
724 (define (walk* v s)
725   (let ((v (walk v s)))
726     (cond
727       ((var? v) v)
728       ((pair? v) (cons (walk* (car v) s)
729                         (walk* (cdr v) s)))
730       (else v))))
731
732 (define (reify-s v s)
733   (let ((v (walk v s)))
734     (cond
735

```

```

736 ((var? v)
737   (let ((n (reify-name (length s))))
738     (cons `(. ,n) s)))
739 ((pair? v) (reify-s (cdr v) (reify-s (car v) s)))
740 (else s)))
741
742 (define (reify-name n)
743   (string->symbol
744     (string-append "_" "." (number->string n))))
745
746 (define (fresh/nf n f)
747   (letrec
748     ((app-f/v*
749      (lambda (n v*)
750        (cond
751          ((zero? n) (apply f (reverse v*)))
752          (else (call/fresh
753                  (lambda (x)
754                    (app-f/v* (- n 1) (cons x v*))))))))))
755     (app-f/v* n '())))
756

```

## 757 REFERENCES

- 758 Claire E. Alvis, Jeremiah J. Willcock, Kyle M. Carter, William E. Byrd, and Daniel P. Friedman. 2011. cKanren: miniKanren  
 759 with Constraints. *Proceedings of the 2011 Workshop on Scheme and Functional Programming*.
- 760 William E. Byrd. 2009. *Relational programming in miniKanren: techniques, applications, and implementations*. Ph.D.  
 761 Dissertation. Bloomington, IN, USA.
- 762 William E. Byrd, Eric Holk, and Daniel P. Friedman. 2012. miniKanren, Live and Untagged: Quine Generation via Relational  
 763 Interpreters (Programming Pearl). In *Proceedings of the 2012 Annual Workshop on Scheme and Functional Programming*  
 764 (*Scheme '12*). ACM, New York, NY, USA, 8–29. <https://doi.org/10.1145/2661103.2661105>
- 765 Daniel P. Friedman, William E. Byrd, Oleg Kiselyov, and Jason Hemann. 2018. *The Reasoned Schemer, Second Edition*. The  
 766 MIT Press, Cambridge, MA, USA.
- 767 Steve Harris and Andy Seaborne. 2013. *SPARQL 1.1 Query Language*. Technical Report. <https://www.w3.org/TR/sparql11-query/>
- 768 Moritz Heidkamp. 2013. persistent-hash-map. (2013). <http://wiki.call-cc.org/eggref/4/persistent-hash-map>
- 769 Jason Hemann and Daniel P. Friedman. 2013.  $\mu$ Kanren: A Minimal Functional Core for Relational Programming. *Proceedings*  
 770 *of the 2013 Workshop on Scheme and Functional Programming*.
- 771 Jason Hemann and Daniel P. Friedman. 2015. A Framework for Extending microKanren with Constraints. *Proceedings of the*  
 772 *2015 Workshop on Scheme and Functional Programming*.
- 773 Jason Hemann, Daniel P. Friedman, William E. Byrd, and Might Matthew. 2016. A Small Embedding of Logic Programming  
 774 with a Simple Complete Search. In *Proceedings of the 12th Symposium on Dynamic Languages (DLS 2016)*. ACM, New  
 775 York, NY, USA, 96–107. <https://doi.org/10.1145/2989225.2989230>
- 776 Frank Manola and Eric Miller. 2004. *RDF Primer*. Technical Report. <https://www.w3.org/TR/2004/REC-rdf-primer-20040210/>
- 777 Mehmet A. Orgun and Wanli Ma. 1994. An Overview of Temporal and Modal Logic Programming. In *Proc. First Int. Conf. on*  
 778 *Temporal Logic - LNAI 827*. Springer-Verlag, 445–479.
- 779 Ian A. Price. 2014. Purely Functional Data Structures in Scheme. (2014). <https://github.com/ijp/pfds/blob/master/hamts.sls>
- 780 Cameron Swords and Daniel Friedman. 2013. rKanren: Guided search in miniKanren. In *Proceedings of the 2013 Workshop on*  
 781 *Scheme and Functional Programming*.
- 782 Aad Versteden and Erika Pauwels. 2016. State-of-the-art Web Applications using Microservices and Linked Data. In  
 783 *Proceedings of the 4th Workshop on Services and Applications over Linked APIs and Data*.
- 784 Aad Versteden and Erika Pauwels. 2018. mu.semte.ch: State-of-the-art Web Applications fuelled by Linked Data. (2018).  
 Presentation at the Extended Semantic Web Conference 2018, Heraklion, Greece.